

## Nauty in Macaulay2

DAVID COOK II

**ABSTRACT.** We introduce a new *Macaulay2* package, *Nauty*, which gives access to powerful methods on graphs provided by the software **nauty** by Brendan McKay. The primary motivation for accessing **nauty** is to determine if two graphs are isomorphic. We also implement methods to generate families of graphs restricted in various ways using tools provided with the software **nauty**.

**INTRODUCTION.** Let  $G$  and  $H$  be two finite, simple, undirected graphs on the common vertex set  $V$  with edge sets  $E(G)$  and  $E(H)$ , respectively. We say that  $G$  and  $H$  are *isomorphic* if there is a bijection  $\varphi$  from  $V$  to itself which preserves edges, that is,  $\{u, v\} \in E(G)$  if and only if  $\{\varphi(u), \varphi(v)\} \in E(H)$ . Determining whether two given graphs are isomorphic is known as the Graph Isomorphism problem.

When Garey and Johnson wrote their classic book [GJ] on the complexity of algorithms, they specified twelve problems of ambiguous complexity, one of which was the Graph Isomorphism problem. Unfortunately, it is still unknown if the Graph Isomorphism problem is **P** or **NP**-complete. Moreover, the problem is of such notoriety that some have even begun referring to a new complexity class, **GI**, of problems which reduce in polynomial time to the Graph Isomorphism problem [J]. Despite this, there exists computer software which is capable of determining whether two graphs are isomorphic in reasonable time. One such piece of software is **nauty** [N] by McKay.

The **nauty** software is written in highly portable *C* and is designed to, above all else, compute whether two graphs are isomorphic. It also includes an extensive family of tools, collectively called **gtools**, to generate systematic modifications of graphs, to generate specific families of graphs, to generate random graphs, to filter a set of graphs for given properties, and to canonically relabel graphs. Most of these features would be beneficial to any computer software that handles graphs.

The package *EdgeIdeals* [FHT] by Francisco, Hoefel, and Van Tuyl implements structures and methods for manipulating graphs (and hypergraphs) within *Macaulay2* [M2], a software system by Grayson and Stillman designed to aid in research of commutative algebra and algebraic geometry. We introduce a new package, *Nauty*, for *Macaulay2*, which provides an interface with **nauty**.<sup>1</sup> Most of the aforementioned tools in **gtools** are accessible through *Nauty*. In particular, the methods of perhaps the greatest interest are `areIsomorphic`, `filterGraphs`, `generateGraphs`, and `generateRandomGraphs`.

The remainder of this note is broken in to two sections: the first describes briefly the theoretical underpinnings of **nauty** and the second gives an example session of using *Nauty* along with a few useful caveats.

**CANONICAL LABELLINGS.** In [M], McKay describes the improved algorithms which he developed to canonically label a graph; these algorithms are the heart of **nauty** and are summarised in [N]. We recall briefly the theoretical ideas which make such algorithms useful.

---

2010 *Mathematics Subject Classification.* 05C25.

*Nauty* version 1.4.1; **nauty** version 2.4r2.

<sup>1</sup>Throughout this note we use **nauty** to refer to the software and *Nauty* to refer to the interface.

Let  $G$  be a finite, simple, undirected graph on the vertex set  $V$ . We call an ordered partition  $\pi = (V_1, \dots, V_m)$  of  $V$  a *colouring* of  $G$  and call  $(G, \pi)$  a *coloured graph*. Given a permutation  $\sigma$  of  $V$ , we define  $\sigma(V_i)$  to be the set  $\{\sigma(v) \mid v \in V_i\}$ ,  $\sigma(\pi)$  to be the colouring  $(\sigma(V_1), \dots, \sigma(V_m))$ , and  $\sigma(G)$  to be the graph on  $V$  with edge set  $E(\sigma(G))$  given by  $\{\sigma(u), \sigma(v)\} \in E(\sigma(G))$  for all  $\{u, v\} \in E(G)$ . If  $\sigma(\pi) = \pi$ , then  $\sigma$  is called *colour-preserving*.

If  $V = [n] = \{1, \dots, n\}$  and  $\pi = (V_1, \dots, V_m)$  is a colouring of  $V$ , then  $c(\pi)$  is the colouring  $(\{1, \dots, |V_1|\}, \{|V_1| + 1, \dots, |V_1| + |V_2|\}, \dots, \{n - |V_m| + 1, \dots, n\})$ . A *canonical labelling map* is a function  $\ell$  from the set of coloured graphs with vertex set  $V$  to the set of graphs with vertex set  $V$  such that, for any coloured graph  $(G, \pi)$ ,  $\ell(G, \pi) = \tau(G)$  for some permutation  $\tau$  with  $\tau(\pi) = c(\pi)$  and  $\ell(\sigma(G), \sigma(\pi)) = \ell(G, \pi)$  for every permutation  $\sigma$  of  $V$ .

**Theorem** ([M, Theorem 2.2], [N, Theorem 1]). *Let  $(G, \pi)$  and  $(H, \rho)$  be coloured graphs on the common finite vertex set  $V$  such that  $\pi$  and  $\rho$  have the same number of vertices in the  $i$ -th colour class, for each  $i$ . Further, let  $\ell$  be a canonical labelling map for graphs with vertex set  $V$ . Then  $\ell(G, \pi) = \ell(H, \rho)$  if and only if  $\sigma(G) = H$  for some colour-preserving permutation  $\sigma$ .*

Initially, **nauty** colours a graph with a single colour and then refines this colouring using a specified vertex invariant. Then, using the above theorem, **nauty** can determine if two graphs are isomorphic by checking if their canonical labellings, after refinement, are the same. The power of **nauty** is that it implements fifteen different vertex invariants (see [N, §9]), each of which is more or less useful depending on the class of graphs being tested.

EXAMPLES. We first load *Nauty*, which automatically loads *EdgeIdeals* for access to the Graph class. The **nauty** software stores graphs in two different string formats, Graph6 and Sparse6 (see [N, §19] for a complete description); both formats are handled by *Nauty* in the method `stringToGraph`. The method `graphToString` always returns the Graph6 string-representation of the graph; e.g., the five-cycle can be represented as “Dhc” and the complete graph on five vertices is represented as “D~{” in the Graph6 string-representation.

```
Macaulay2, version 1.4
with packages: ConwayPolynomials, Elimination, IntegralClosure, LLLBases,
               PrimaryDecomposition, ReesAlgebra, TangentCone

i1 : needsPackage "Nauty";
--loading configuration for package "Nauty" from file ../Macaulay2/init-Nauty.m2
i2 : R = QQ[a..e];
i3 : graphToString cycle R
o3 = Dhc
i4 : graphToString completeGraph R
o4 = D~{
i5 : edges stringToGraph("Dhc", R)
o5 = {{a, b}, {b, c}, {c, d}, {a, e}, {d, e}}
o5 : List
```

**Caveat.** Due to the way graphs are constructed in *EdgeIdeals*, the conversion process from string format to graph format is costly. Thus, when many consecutive manipulations will be done through *Nauty*, we recommended that the graphs be left in string format until it is necessary to convert back to graph format.

The most powerful feature of *Nauty* is determining if two graphs are isomorphic. We can demonstrate it for two given graphs with `areIsomorphic`; further we can reduce a list of graphs to be pairwise non-isomorphic with `removeIsomorphs`. We demonstrate the latter on a list of  $5! = 120$  different labellings of the five-cycle, which is represented as “Dhc” in the Graph6 string-representation.

```
i6 : G = graph {{a, c}, {c, e}, {e, b}, {b, d}, {d, a}};
i7 : areIsomorphic(cycle R, G)
o7 = true
i8 : removeIsomorphs apply(permutations gens R,
    P -> graphToString graph apply(5, i-> {P_i, P_((i+1)%5)}))
o8 = {Dhc}
o8 : List
```

*Nauty* also includes methods for generating all graphs on a given number of vertices, possibly with restrictions to simple properties such as the number of edges, with the method `generateGraphs`. We generate all graphs with between one and nine vertices and verify the counts with the Online Encyclopedia of Integer Sequences [S, A000088].

**Caveat.** As *nauty* does not handle graphs with zero vertices, *Nauty* will throw an error when graphs with zero vertices are requested or encountered.

```
i9 : A000088 = apply(1..9, n -> #generateGraphs n)
o9 = (1, 2, 4, 11, 34, 156, 1044, 12346, 274668)
o9 : Sequence
```

We can also select all graphs in a list with a given property. To do this, we first build a filter with the method `buildGraphFilter` and then use the method `filterGraphs` to select all graphs in the list which pass the filter; see the documentation for `buildGraphFilter` to see all the possible properties which can be filtered for. First we generate all bipartite graphs with between one and twelve vertices. We then filter this list for forests, i.e., graphs without cycles. We verify the counts with [S, A005195].

```
i10 : B = apply(1..12, n -> generateGraphs(n, OnlyBipartite => true));
i11 : forestsOnly = buildGraphFilter {"NumCycles" => 0};
i12 : A005195 = apply(B, graphs -> #filterGraphs(graphs, forestsOnly))
o12 = (1, 2, 3, 6, 10, 20, 37, 76, 153, 329, 710, 1601)
o12 : Sequence
```

**Caveat.** *Nauty* uses the following definition of  $k$ -connectedness: a 0-connected graph is a disconnected graph and, for  $k > 0$ , a  $k$ -connected graph is a graph that can be disconnected by removing  $k$  vertices but not by removing  $k - 1$  vertices.

Thus, if we wish to select only connected graphs, then we must create a filter for graphs which are *not* 0-connected. Alternatively, the method `generateGraphs` has the option `OnlyConnected` which forces only connected graphs to be returned. We demonstrate the former by filtering the list of bipartite graphs for trees, i.e., connected graphs without cycles. We verify the counts with [S, A000055].

```
i13 : treesOnly = buildGraphFilter {"NumCycles" => 0,
    "Connectivity" => 0, "NegateConnectivity" => true};
i14 : A000055 = apply(B, graphs -> #filterGraphs(graphs, treesOnly))
o14 = (1, 1, 1, 2, 3, 6, 11, 23, 47, 106, 235, 551)
```

o14 : Sequence

Last, when testing conjectures — especially when dealing with graphs on many vertices — it is nice to be able to generate large lists of random graphs. *EdgeIdeals* provides the method `randomGraph`, but this method only generates graphs with a specified number of edges and, moreover, generates only one graph at a time. *Nauty* provides the method `generateRandomGraphs`, which randomly generates multiple graphs on a fixed number of vertices with a specified edge probability.

Erdős and Rényi show in [ER, Theorem 1] that a random graph on  $n$  vertices with edge probability  $\frac{(1+\varepsilon)\log n}{n}$  is almost always connected while a graph on  $n$  vertices with edge probability  $\frac{(1-\varepsilon)\log n}{n}$  is almost never connected, at least as  $n$  tends to infinity. We demonstrate this property by generating 100 random graphs on  $n$  vertices where  $\varepsilon$  is 1 in the first case and  $\frac{1}{2}$  in the second.

```
i15 : connected = buildGraphFilter {"Connectivity" => 0,
    "NegateConnectivity" => true};
i16 : prob = n -> log(n)/n;
i17 : apply(2..18, n-> #filterGraphs(
    generateRandomGraphs(n, 100, 2*(prob n)), connected))
o17 = (65, 84, 82, 95, 94, 97, 98, 99, 95, 99, 97, 95, 96, 98, 98, 99, 98)
o17 : Sequence
i18 : apply(2..18, n-> #filterGraphs(
    generateRandomGraphs(n, 100, (prob n)/2), connected))
o18 = (18, 12, 7, 7, 5, 3, 0, 6, 1, 1, 6, 1, 0, 2, 1, 2, 0)
o18 : Sequence
```

ACKNOWLEDGEMENTS. We thank the anonymous referee for helpful comments, especially regarding the package documentation. We also thank the Editorial Board of the journal for many instructive comments on documentation and coding in *Macaulay2*.

#### REFERENCES.

- [ER] P. Erdős and A. Rényi, *On random graphs. I*, Publ. Math. Debrecen **6** (1959), 290 – 297.
- [FHT] C.A. Francisco, A. Hoefel, and A. Van Tuyl, *EdgeIdeals: a package for (hyper)graphs*, JSAG **1** (2009), 1 – 4.
- [GJ] M.R. Garey and D.S. Johnson, *Computers and intractability: A guide to the theory of NP-completeness*, W. H. Freeman and Co., San Francisco, Calif., 1979.
- [J] D.S. Johnson, *The NP-completeness column*, ACM Trans. Algorithms **1** (2005), no. 1, 160 – 176.
- [M2] D.R. Grayson and M.E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at [www.math.uiuc.edu/Macaulay2/](http://www.math.uiuc.edu/Macaulay2/).
- [M] B.D. McKay, *Practical graph isomorphism*, Proceedings of the Tenth Manitoba Conference on Numerical Mathematics and Computing, Vol. I (Winnipeg, Man., 1980), 1981, pp. 45 – 87.
- [N] ———, *nauty User's Guide (Version 2.4)*, available at <http://cs.anu.edu.au/~bdm/nauty/>.
- [S] N.J.A. Sloane, *The On-Line Encyclopedia of Integer Sequences*, available at <http://oeis.org>.

RECEIVED : 2010-10-29

REVISED : 2011-02-23

ACCEPTED : 2011-04-20

[dcook@ms.uky.edu](mailto:dcook@ms.uky.edu) : Department of Mathematics, University of Kentucky, 715 Patterson Office Tower, Lexington, KY 40506-0027, USA.